


9/3.3

Tours de Hanoï

Ce jeu de réflexion met en œuvre des tours de diamètres différents empilées dans l'ordre le plus petit diamètre en haut, le plus grand en bas.

Ces tours sont empilées dans une des trois zones possibles. Le but du jeu consiste à reconstituer l'empilement de départ sur l'emplacement opposé en un minimum de coups. Le nombre de tours au départ est choisi par le joueur.

L'organigramme du programme est le suivant :



```

1000 MODE 1:PRINT"Tours de Hanoi"
1010 '=====
1020 'Initialisation
1030 '=====
1040 INPUT "Nombre de disques (entre 3 et 7)";ND
1050 IF ND>7 OR ND<3 THEN GOTO 1040
1060 '
1070 FOR J=0 TO 3
1080   FOR I=0 TO 8
1090     T(I,J)=0
1100   NEXT I
1110 NEXT J
1120 FOR I=1 TO ND
1130   T(I,1)=I
1140 NEXT I
1150 '=====
1160 'Partie
1170 '=====
1180 CLS
1190 FOR J=1 TO 3
1200   FOR I=1 TO ND
1210     IF T(I,J)=0 THEN 1270
1220     A=(J-1)*95+50-5*(8+T(I,J)-ND)
1230     B=150+12*(ND-I+1)
1240     C=(J-1)*95+60+5*(8+T(I,J)-ND)
1250     D=B+12
1260     MOVE A,B:DRAW C,B:DRAW C,D:DRAW A,D:DRAW A,B
1270   NEXT I
1280 NEXT J
1290 '
1300 INPUT"Tour de depart ";TD
1310 INPUT"Tour d'arrivee ";TA
1320 '
1330 IF TA>3 OR TD>3 OR TA<1 OR TD<1 THEN OK=1:GOTO 1420
1340 I=ND
1350 IF T(I,TA)<>0 THEN I=I-1:GOTO 1350
1360 J=ND
1370 IF T(J,TD)<>0 THEN J=J-1:GOTO 1370
1380 J=J+1:OK=1
1390 IF I=ND AND J<=ND THEN OK=-1:GOTO 1420
1400 IF J>ND THEN OK=1:GOTO 1420
1410 IF T(I+1,TA)>T(J,TD) THEN OK=-1
1420 IF OK=1 THEN PRINT"Deplacement non accepte":GOTO 1300
1430 '
1440 T(I,TA)=T(J,TD):T(J,TD)=0
1450 '
1460 BP=0
1470 FOR I=1 TO ND
1480   IF T(I,3)=I THEN BP=BP+1
1490 NEXT I
1500 NC=NC+1
1510 IF BP<>ND THEN 1180
1520 '=====
1530 'Fin de partie
1540 '=====
1550 PRINT"Tours de Hanoi deplacees en ";NC;"coups"

```

Lignes 1000 à 1140 : Initialisation des tours dans leur position initiale.

Lignes 1150 à 1510 : Jeu

Saisie du déplacement : Lignes 1300 et 1310

Test de cohérence : Lignes 1330 à 1410

Comptabilisation
du déplacement : Lignes 1460 à 1500

Lignes 1520 à 1550 : Affichage des résultats

